



# Optimisation sur un domaine fermé

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue à l'automne 2006. Je vous présente ici quelques exemples pour animer en classe des discussions à propos de l'optimisation sur des domaines fermés. Le premier exemple (Exemple 0) expose les étapes analytiques requises dans la recherche des extrema globaux sur un domaine borné fermé. Les autres exemples (1 à 5) présentent des surfaces ayant leurs extrema globaux (pas nécessairement unique) sur la frontière et/ou à l'intérieur du domaine borné fermé. Le dernier exemple propose aussi une résolution avec des macro-commandes de l'extension simplex puisque la fonction  $f$  à optimiser est linéaire avec contraintes linéaires.

Bonne lecture à tous !

\* Ce document Maple est exécutable avec la version 2020.2

*Théorème des valeurs extrêmes pour les fonctions de deux variables.*

Si  $f$  est continue sur un ensemble borné fermé de  $D \subseteq \mathbb{R}^2$ , alors  $f$  atteint un maximum absolu  $f(x_1, y_1)$  et un minimum absolu  $f(x_2, y_2)$ , en des points  $(x_1, y_1)$  et  $(x_2, y_2)$  de  $D$ .

Pour trouver les valeurs maximale et minimale absolues sur un domaine  $D$  (borné et fermé), nous devons:

1. calculer les valeurs de  $f$  aux points stationnaires de  $f$  dans  $D$
2. calculer les valeurs extrêmes de  $f$  sur la frontière (bord) de  $D$ .
3. dans la liste de ces valeurs, la plus grande valeur est la valeur maximale absolue tandis que la plus petite valeur est la valeur minimale absolue.

Remarque: Cette démarche est similaire à la démarche d'un problème d'optimisation sur un intervalle fermé d'une fonction réelle à une variable réelle.

## Initialisation

```
> restart;  
> with(plots,display,pointplot3d,polygonplot3d,textplot3d,spacecurve,  
  setoptions3d):  
  with(plottools,disk,transform):  
  with(VectorCalculus,Gradient,Vector):  
  
> setoptions3d(size=[450,450],labels=[x,y,z],axesfont=[times,roman,8],  
  axis=[color="Niagara 4"],  
  labelfont=[times,roman,12],titlefont=[times,roman,14]);
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimant les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de  
  composition étendue
```

```
Typesetting:-Settings(striptrailing=true);
                                extended
                                false
```

(1.1)

```
> Digits:=20;
interface( displayprecision=5 );
                                Digits := 20
                                5
```

(1.2)

## Exemple 0


Analyse Stewart, Exemple 7 page 817 :

Quelles sont les valeurs extrêmes absolues (extrema globaux) de la fonction  $f(x, y) = x^2 - 2xy + 2y$  sur le rectangle  $D = \{(x, y) \mid 0 \leq x \leq 3, 0 \leq y \leq 2\}$  ?

```
> f:=(x,y)->x^2-2*x*y+2*y;
                                f := (x, y) ↦ x2 - 2·y·x + 2·y
```

(2.1)

```
> Surface:=plot3d([x,y,f(x,y)],x=0..3,y=0..2,grid=[20,20]):
Domaine:=plot3d([x,y,0],x=0..3,y=0..2,grid=[20,20],style=surface,
color="HTML 12"):
display([Surface,Domaine],axes=normal,orientation=[-20,70,0]);
```



Cherchons les points stationnaires de  $f$ . Obtenons  $\frac{\partial f}{\partial x}$  et  $\frac{\partial f}{\partial y}$ .

```
> 'diff'(f(x,y),x)=diff(f(x,y),x);
'diff'(f(x,y),y)=diff(f(x,y),y);
                                ∂
                                ∂x (x2 - 2yx + 2y) = 2x - 2y
```

$$\frac{\partial}{\partial y} (x^2 - 2yx + 2y) = -2x + 2 \quad (2.2)$$

Réolvons le système  $\begin{cases} 2x - 2y = 0 \\ -2x + 2 = 0 \end{cases}$


```
> solve({diff(f(x,y),x),diff(f(x,y),y)});
           {x=1,y=1}
(2.3)
```

```
> f(1,1);
           1
(2.4)
```

Le seul point stationnaire est (1,1) et il est à l'intérieur de D (pourquoi ?). On retient donc ce point. La valeur que  $f$  atteint en ce point est 1. Nul besoin de tester ce point car nous sommes à la recherche des extremum globaux  $f$ .

Illustrons ensuite le domaine D dans le plan cartésien.

```
> L1:=textplot3d([1.5,0,0,typeset(L[1])],align=left):
L2:=textplot3d([3,1,0,typeset(L[2])],align=below):
L3:=textplot3d([1.5,2.01,0.01,typeset(L[3])],align=right):
L4:=textplot3d([0,1,0.02,typeset(L[4])],align=above):
CoordP1:=textplot3d([0,0,0,typeset("(0,0,0)"),align=left):
CoordP2:=textplot3d([0,2,0.01,typeset("(0,2,0)"),align=above):
CoordP3:=textplot3d([3,2,0,typeset("(3,2,0)"),align=right):
CoordP4:=textplot3d([3,0,0,typeset("(3,0,0)"),align=below):
display(Domaine,style=surface,L|(1..4),CoordP|(1..4),
        orientation=[-20,70,0],
        tickmarks=[0,0,0],
        labels=[``,``,``],
        color="Niagara 4",
        axes=normal,
        view=[0..3,0..2.2,0..1],
        scaling=constrained,size=[400,400]);
```



Le domaine  $D$  possède quatre frontières  $L_1$ ,  $L_2$ ,  $L_3$ , et  $L_4$ . Analysons  $f$  sur chacune de ces frontières. L'analyse sur chaque frontière va consister, dans cet exemple, à analyser une fonction d'une variable sur un intervalle fermé. Dans les autres exemples, nous ferons intervenir l'analyse habituelle (dérivation et valeurs critiques) d'une fonction d'une variable sur un intervalle fermé.

- Sur  $L_1, y = 0$  alors  $f(x, 0) = x^2$  avec  $0 \leq x \leq 3$ . Il s'agit d'une fonction strictement croissante de  $x$ , son minimum est  $f(0, 0) = 0$  et son maximum est  $f(3, 0) = 9$ .
- Sur  $L_2, x = 3$  alors  $f(3, y) = 9 - 4y$  avec  $0 \leq y \leq 2$ . Il s'agit d'une fonction strictement décroissante de  $y$ , son maximum est  $f(3, 0) = 9$  et son minimum est  $f(3, 2) = 1$ .
- Sur  $L_3, y = 2$  alors  $f(x, 2) = x^2 - 4x + 4$  avec  $0 \leq x \leq 3$ . Il est facile de voir que la règle  $f(x)$  est un carré parfait  $f(x, 2) = (x - 2)^2$  sinon, il aurait fallu transformer la forme générale sous la forme canonique pour mieux analyser la forme quadratique. Comme forme quadratique où  $a > 0$ , le minimum est au sommet en  $x = 2$ , donc  $f(2, 2) = 0$  est le maximum à la borne  $x = 0$  d'où  $f(0, 2) = 4$  (à l'autre borne  $x = 3, f(3, 2) = 1$  est une valeur inférieure).
- Sur  $L_4, x = 0$  alors  $f(0, y) = 2y$  avec  $0 \leq y \leq 2$ . Il s'agit d'une fonction croissante, donc le minimum est  $f(0, 0) = 0$  et le maximum est  $f(0, 2) = 4$ .

Comparons les valeurs obtenues.

#### Tableau comparatif

(1,1) pt critique	(0, 0)	(3, 0)	(3, 2)	(2, 2)	(0, 2)
1	0	9	1	0	4

En conclusion, le maximum absolu de  $f$  sur  $D$  est 9 et est atteint en  $(3,0)$ . Le minimum absolu est 0 et est atteint en  $(0,0)$  et en  $(2,2)$ .

**Remarque :** La solution précédente est une indication quant aux étapes à suivre dans la détermination des extrema globaux. Dans cet exemple, le recours à Maple pour les points stationnaires ou pour l'analyse sur les frontières n'a pas été indispensable.

## Exemple 1


Soit  $f$  définie par  $f(x,y) = \sqrt{x^2 + y^2}$ . Le domaine de  $f$  est  $\mathbb{R}^2$ . Optimiser  $f$  sur le carré  $D = \{(x,y) \mid -3 \leq x \leq 3, -3 \leq y \leq 3\}$

```
> f:=(x,y)->sqrt(x^2+y^2);
```

$$f := (x,y) \mapsto \sqrt{y^2 + x^2}$$


(3.1)

```
> Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-5..5,shading=
zhue,transparency=0.4,color="Niagara 4"):
Domaine:=plot3d([x,y,0],x=-5..5,y=-5..5,grid=[20,20],style=wireframe,
color="Niagara 16"):
Domaine_optimisation:=polygonplot3d([[ -3,-3,0],[ -3,3,0],[3,3,0],[3,
-3,0],[ -3,-3,0]],grid=[20,20],linestyle=solid,color="Niagara 14"):
display([Surface,Domaine_optimisation,Domaine],axes=framed,
orientation=[70,60,0]);
```



Nous aurions pu également tracer la surface  $f$  sur un domaine de tracé circulaire au lieu d'un domaine de tracé rectangulaire (qui doit inclure aussi le domaine d'optimisation tout de même).

```
> Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-x^2),
shading=zhue,transparency=0.4,color="Niagara 4"):
display([Surface,Domaine_optimisation,Domaine],axes=framed,
orientation=[70,60,0]);
```



Recherchons les points stationnaires (même si on peut le "voir" sur le tracé de la surface).

**> Nabra:=Gradient(f(x,y),[x,y]);**

$$\nabla := \left( \frac{x}{\sqrt{x^2 + y^2}} \right) \bar{e}_x + \left( \frac{y}{\sqrt{x^2 + y^2}} \right) \bar{e}_y \quad (3.2)$$

Il est évident que, pour cette fonction f, nous n'avons aucun couple (x, y) pour lequel le gradient s'annule.

**> solve({Nabra[1]=0,Nabra[2]=0},{x,y});**

D'autre part, le gradient de f existe toujours sauf en (0, 0). Le gradient de f ne s'annulant jamais, le seul point stationnaire est (0, 0). De plus, en ce point (0, 0), quelque soit la direction des accroissements  $h \cos(\alpha)$  à la variable x et  $h \sin(\alpha)$  à la variable y,  $z = f(x, y)$  croît toujours. Nous déduisons immédiatement que 0 est le minimum absolu et qu'il est atteint en (0, 0).

Nous ne ferons pas immédiatement le calcul des valeurs de f sur l'intérieur ni sur chaque frontière du domaine d'optimisation. Au fur et à mesure du développement, nous allons que retenir les points stationnaires intérieurs et les points critiques sur chaque frontière. Et à la fin, dans un tableau comparatif, nous allons mettre en évidence les extrema globaux recherchés.

Le point (0, 0) est donc à retenir. (La notation d'ensemble va nous permettre d'éviter les redondances.)

**> ValeursRetenues:={ [0,0]};**

$$ValeursRetenues := \{ [0,0] \} \quad (3.3)$$

Analysons les frontières. Il y en a quatre.

Sur  $L_1$  où  $y = -3$ : analysons la fonction f pour  $-3 \leq x \leq 3$ . Sur  $L_1$  la fonction f devient

**> f(x,-3);**

$$\sqrt{x^2 + 9} \quad (3.4)$$

Trouvons les valeurs critiques de f sur cette frontière. Les bornes de l'intervalle  $[-3, 3]$  sont d'office des valeurs critiques.



Dans un tableau, comparons les valeurs de la fonction avec les points retenus.

```
> Valeurs_C:=op(ValeursRetenues);
    Valeurs_C := [-3, -3], [-3, 0], [-3, 3], [0, -3], [0, 0], [0, 3], [3, -3], [3, 0], [3, 3]      (3.16)
```

Calculs des valeurs de la fonction f en ces points.

```
> printf("\n%18s \n","Tableau comparatif des valeurs de f aux points
retenus");
printf(`\n
    Points retenus      |   Valeurs de f   |
    Approx. de f`);
printf(`\n
=====
=====`);
seq(printf(`\n %35a      |%13a      |      %5g`, [Valeurs_C][k],f(op(1,op
(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),
    evalf(f(op(1,op(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))
))),k=1..nops([Valeurs_C]));
#seq(printf(`\n %35a      | %10a |      %5g`, [Valeurs_C][k],f(op(1,op
(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),op(2,op(k,[Valeurs_C]))),k=
1..nops([Valeurs_C]));
```

Tableau comparatif des valeurs de f aux points retenus

Points retenus	Valeurs de f	Approx. de f
[-3, -3]	$3 \cdot 2^{(1/2)}$	4.24264
[-3, 0]	3	3
[-3, 3]	$3 \cdot 2^{(1/2)}$	4.24264
[0, -3]	3	3
[0, 0]	0	0
[0, 3]	3	3
[3, -3]	$3 \cdot 2^{(1/2)}$	4.24264
[3, 0]	3	3
[3, 3]	$3 \cdot 2^{(1/2)}$	4.24264

En conclusion, le maximum absolu de f sur D est  $3\sqrt{2}$  et est atteint en les points  $(-3,-3)$ ,  $(3,-3)$ ,  $(-3,3)$  et  $(3,3)$ . Le minimum absolu est 0 et est atteint en  $(0,0)$ .

Illustrons les résultats obtenus.


```
> Projection:=spacecurve([[-3,-3,f(-3,-3)],[-3,3,f(-3,3)],[3,3,f(3,3)],
[3,-3,f(3,-3)],[-3,-3,f(-3,-3)]],
    thickness=2,
    linestyle=2,
    color="Niagara 1"):
Lieu1:=spacecurve([x,3,f(x,3)],x=-3..3,color="Niagara 2",thickness=1)
:
Lieu2:=spacecurve([x,-3,f(x,-3)],x=-3..3,color="Niagara 2",thickness=
1):
Lieu3:=spacecurve([3,y,f(3,y)],y=-3..3,color="Niagara 2",thickness=1)
```



```

:
Lieu4:=spacecurve([-3,y,f(-3,y)],y=-3..3,color="Niagara 2",thickness=
1):
Segment1:=pointplot3d([[-3,-3,0],[-3,-3,f(-3,-3)]],style=line,color=
"Niagara 2",linestyle=3,thickness=1):
Segment2:=pointplot3d([[-3,3,0],[-3,3,f(-3,3)]],style=line,color=
"Niagara 2",linestyle=3,thickness=1):
Segment3:=pointplot3d([[3,-3,0],[3,-3,f(3,-3)]],style=line,color=
"Niagara 2",linestyle=3,thickness=1):
Segment4:=pointplot3d([[3,3,0],[3,3,f(3,3)]],style=line,color=
"Niagara 2",linestyle=3,thickness=1):
Segment5:=pointplot3d([[-3,0,0],[-3,0,f(-3,0)]],style=line,color=
"Niagara 1",linestyle=2,thickness=1):
Segment6:=pointplot3d([[0,3,0],[0,3,f(0,3)]],style=line,color=
"Niagara 2",linestyle=2,thickness=1):
Segment7:=pointplot3d([[0,-3,0],[0,-3,f(0,-3)]],style=line,color=
"Niagara 1",linestyle=2,thickness=1):
Segment8:=pointplot3d([[3,0,0],[3,0,f(3,0)]],style=line,color=
"Niagara 2",linestyle=2,thickness=1):
Points:=pointplot3d([[0,0,0],[-3,-3,f(-3,-3)],[3,-3,f(3,-3)],[-3,3,f
(-3,3)],[3,3,f(3,3)]],symbol=solidsphere,symbolsize=18,color="Niagara
4"):
AutresPoints:=pointplot3d([[-3,0,f(-3,0)],[0,-3,f(0,-3)],[0,3,f(0,3)
],[3,0,f(3,0)]],symbol=solidsphere,symbolsize=18,color="Niagara 4"):
A:=display([Surface,Domaine_optimisation,Domaine,Projection,Lieu|(1.
.4)],axes=frame,view=[-4.3..4.3,-4.3..4.3,0..f(3,3)],orientation=[55,
65,0]):
SurfaceBis:=plot3d([x,y,f(x,y)],x=-3..3,y=-3..3,shading=zhue,
transparency=0.4,color="Niagara 4"):
B:=display([SurfaceBis,Domaine_optimisation,Domaine,Projection,
Points,Lieu|(1..4)],Segment|(1..4),axes=frame,view=[-4.3..4.3,-4.3.
.4.3,0..f(3,3)],orientation=[55,65,0]):
C:=display([Domaine_optimisation,Domaine,Projection,Points,
AutresPoints,Lieu|(1..4)],Segment|(1..8),axes=frame,view=[-4.3.
.4.3,-4.3..4.3,0..f(3,3)],orientation=[55,65,0]):
display(Matrix(1,3,[A,B,C]));

```




## Exemple 2

Soit à optimiser  $f(x,y) = 20 - (x-1)^2 - (y-1)^2$  sur le domaine  $D = \{(x,y) \mid 0 \leq x \leq 4, 0 \leq y \leq 4-x\}$ .

```
> f:=(x,y)->20-(x-1)^2-(y-1)^2;
      f:=(x,y) ↦ 20 - (x-1)2 - (y-1)2 (4.1)
```

```
> Surface:=plot3d([x,y,f(x,y)],x=0..4,y=0..4,grid=[20,20],shading=zhue)
:
Domaine:=plot3d([x,y,0],x=0..5,y=0..5,grid=[20,20],style=wireframe,
color="Niagara 16"):
Domaine_optimisation:=plot3d([x,y,0],x=0..4,y=4-x..0,grid=[25,25],
linestyle=solid,style=patchnogrid,color="HTML 26"):
display([Surface,Domaine,Domaine_optimisation],axes=normal,
orientation=[20,70,0]);
```



Déterminons si  $f$  possède des points stationnaires à l'intérieur du domaine d'optimisation.

$$\begin{aligned} > \text{Nabla} := \text{Gradient}(f(x,y), [x,y]); \\ & \qquad \qquad \qquad \nabla := (-2x+2)\bar{e}_x + (-2y+2)\bar{e}_y \end{aligned} \quad (4.2)$$

$$\begin{aligned} > \text{solve}(\{\text{Nabla}[1]=0, \text{Nabla}[2]=0\}, \{x,y\}); \\ & \qquad \qquad \qquad \{x=1, y=1\} \end{aligned} \quad (4.3)$$

Effectivement, le point  $(1, 1)$  est un point intérieur (quand  $x = 1$ , l'ordonnée  $1 < y = 4 - x = 3$ ).

Au fur et à mesure du développement, nous allons que retenir les points stationnaires intérieurs et les points critiques sur chaque frontière. Et à la fin, dans un tableau comparatif, nous allons mettre en évidence les extremum globaux recherchés.

Le point stationnaire  $(1, 1)$  est à retenir.

$$\begin{aligned} > \text{ValeursRetenues} := \{[1,1]\}; \\ & \qquad \qquad \qquad \text{ValeursRetenues} := \{[1,1]\} \end{aligned} \quad (4.4)$$

Analysons les frontières. Il y en a trois.

Sur  $L_1$  où  $y = 0$ : analysons la fonction pour  $0 \leq x \leq 4$ .

$$\begin{aligned} > f(x,0); \\ & \qquad \qquad \qquad 19 - (x-1)^2 \end{aligned} \quad (4.5)$$

Trouvons les valeurs critiques pour  $0 < x < 4$ .

$$\begin{aligned} > \text{solve}(\text{diff}((4.5),x)=0, \{x\}); \\ & \qquad \qquad \qquad \{x=1\} \end{aligned} \quad (4.6)$$

Alors, sur  $L_1$ , on retient les points  $(1, 0)$ ,  $(0, 0)$  et  $(4, 0)$ .

$$\begin{aligned} > \text{ValeursRetenues} := \text{ValeursRetenues union } \{[1,0],[0,0],[4,0]\}; \\ & \qquad \qquad \qquad \text{ValeursRetenues} := \{[0,0], [1,0], [1,1], [4,0]\} \end{aligned} \quad (4.7)$$

Sur  $L_2$  où  $x = 0$ : analysons la fonction pour  $0 \leq y \leq 4$ .

$$\begin{aligned} > f(0,y); \\ & \qquad \qquad \qquad 19 - (y-1)^2 \end{aligned} \quad (4.8)$$

Trouvons les valeurs critiques pour  $0 < y < 4$ .

$$\begin{aligned} > \text{solve}(\text{diff}((4.8),y)=0, \{y\}); \\ & \qquad \qquad \qquad \{y=1\} \end{aligned} \quad (4.9)$$

Alors, sur  $L_2$ , on retient les points  $(0, 1)$ ,  $(0, 0)$  et  $(0, 4)$ .

$$\begin{aligned} > \text{ValeursRetenues} := \text{ValeursRetenues union } \{[0,1],[0,0],[0,4]\}; \\ & \qquad \qquad \qquad \text{ValeursRetenues} := \{[0,0], [0,1], [0,4], [1,0], [1,1], [4,0]\} \end{aligned} \quad (4.10)$$

Sur  $L_3$  où  $y = 4 - x$ : analysons la fonction pour  $0 \leq x \leq 4$ .

$$\begin{aligned} > f(x,4-x); \\ & \qquad \qquad \qquad 20 - (x-1)^2 - (3-x)^2 \end{aligned} \quad (4.11)$$

Trouvons les valeurs critiques pour  $0 < x < 4$ .

$$\begin{aligned} > \text{solve}(\text{diff}((4.11),x)=0, \{x\}); \\ & \qquad \qquad \qquad \{x=2\} \end{aligned} \quad (4.12)$$

Alors, sur  $L_3$ , on retient les points  $(2, 2)$ ,  $(0, 4)$  et  $(4, 0)$ .

```
> ValeursRetenues:=ValeursRetenues union {[2,2],[0,4],[4,0]};
      ValeursRetenues := {[0,0],[0,1],[0,4],[1,0],[1,1],[2,2],[4,0]} (4.13)
```

Dans un tableau, comparons les valeurs de la fonction avec les points retenus. La notation d'ensemble a permis d'éviter les redondances.

```
> Valeurs_C:=op(ValeursRetenues);
      Valeurs_C := [0,0],[0,1],[0,4],[1,0],[1,1],[2,2],[4,0] (4.14)
```

Calculs des valeurs de la fonction  $f$  en ces points.

```
> printf("\n%18s \n","Tableau comparatif des valeurs de f aux points
retenus");
printf(`\n
      Points retenus      |   Valeurs de f   |
      Approx. de f`);
printf(`\n
      =====
      =====`);
seq(printf(`\n %35a      |%13a      |      %5g`, [Valeurs_C][k],f(op(1,op
(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),
      evalf(f(op(1,op(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))))
),k=1..nops([Valeurs_C]));
#seq(printf(`\n %35a      | %10a |      %5g`, [Valeurs_C][k],f(op(1,op
(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),op(2,op(k,[Valeurs_C]))),k=
1..nops([Valeurs_C]));
```

Tableau comparatif des valeurs de  $f$  aux points retenus

Points retenus	Valeurs de $f$	Approx. de $f$
[0, 0]	18	18
[0, 1]	19	19
[0, 4]	10	10
[1, 0]	19	19
[1, 1]	20	20
[2, 2]	18	18
[4, 0]	10	10

En conclusion, le maximum absolu de  $f$  sur  $D$  est 20 et est atteint en  $(1, 1)$  (point intérieur). Le minimum absolu est 10 et est atteint en  $(4, 0)$   $(0, 4)$  sur les bords.


Illustrons les résultats obtenus.

```
> ProjectionHaut:=spacecurve([[0,0,f(2,2)],[4,0,f(2,2)],[0,4,f(2,2)],
[0,0,f(2,2)]],
      thickness=2,linestyle=solid,
      linestyle=2,
      color="Niagara 1"):
ProjectionBas:=spacecurve([[0,0,f(4,0)],[4,0,f(4,0)],[0,4,f(4,0)],[0,
0,f(4,0)]],
      thickness=2,
```

```

        linestyle=2,
        color="Niagara 1"):
Segment0:=pointplot3d([[1,1,0],[1,1,f(1,1)]],style=line,thickness=1,
        linestyle=3,
        color="Niagara 2"):
Lieu1:=spacecurve([x,0,f(x,0)],x=0..4,color="Niagara 2",thickness=1):
Lieu2:=spacecurve([0,4-x,f(0,4-x)],x=0..4,color="Niagara 2",
thickness=1):
Lieu3:=spacecurve([x,4-x,f(x,4-x)],x=0..4,color="Niagara 2",
thickness=1):
Segment1:=spacecurve([[4,0,0],[4,0,f(4,0)]],color="Niagara 2",
linestyle=3,thickness=1):
Segment2:=spacecurve([[0,4,0],[0,4,f(0,4)]],color="Niagara 2",
linestyle=3,thickness=1):
Segment3:=spacecurve([[0,0,0],[0,0,0],[0,0,f(2,2)]],color="Niagara
2",linestyle=2,thickness=1):
Segment4:=spacecurve([[4,0,0],[4,0,f(2,2)]],color="Niagara 1",
linestyle=2,thickness=1):
Segment5:=spacecurve([[0,4,0],[0,4,f(2,2)]],color="Niagara 1",
linestyle=2,thickness=1):
Segment6:=spacecurve([[0,1,0],[0,1,f(0,1)]],color="Niagara 1",
linestyle=2,thickness=1):
Segment7:=spacecurve([[1,0,0],[1,0,f(1,0)]],color="Niagara 1",
linestyle=2,thickness=1):
Segment8:=spacecurve([[2,2,0],[2,2,f(2,2)]],color="Niagara 1",
linestyle=2,thickness=1):
Points:=pointplot3d([[1,1,f(1,1)],[1,0,f(1,0)],[0,0,f(0,0)],[4,0,f(4,
0)],[0,1,f(0,1)],[0,4,f(0,4)],[2,2,f(2,2)]],symbol=solidsphere,
symbolsize=12,color="Niagara 2"):
A:=display([Surface,Domaine_optimisation,Domaine,ProjectionHaut,
ProjectionBas,Points,Lieu| |($1..3),Segment| |($0..8)],size=[500,600],
orientation=[20,70,0]):
B:=display([Domaine_optimisation,Domaine,ProjectionHaut,
ProjectionBas,Points,Lieu| |($1..3),Segment| |($0..8)],orientation=
[-45,85,-10],view=[0..4,0..4,0..f(1,1)]):
display(Matrix(1,2,[A,B]),axes=framed);

```



### Exemple 3

Optimiser  $f(x, y) = x + y + 22$  sur le disque  $x^2 + y^2 \leq 1$ .

#### 3.1 Méthode directe

```
> f:=(x,y)->x+y+22;
```

$$f := (x, y) \mapsto x + y + 22$$

(5.1.1)


```
> Surface:=plot3d([x,y,f(x,y)],x=-2..2,y=-2..2,shading=zhue,grid=[20,20]):
```

```
Domaine:=plot3d([x,y,0],x=-2..2,y=-2..2,grid=[20,20],style=wireframe,color="Niagara 16");
```

```
Domaine_optimisation:=plot3d([x,y,0],x=-1..1,y=-sqrt(1-x^2)..sqrt(1-x^2),style=patchnograd,color="Dalton 2");
```

```
Bord:=spacecurve([cos(t),sin(t),0],t=0..2*Pi,color="Dalton 2",thickness=2):
```

```
display([Surface,Domaine,Domaine_optimisation,Bord],axes=framed,orientation=[70,60,0]);
```



Déterminons si  $f$  possède des points critiques à l'intérieur du domaine d'optimisation.

$$\begin{aligned} > \text{Nabla} := \text{Gradient}(f(x,y), [x,y]); \\ & \qquad \qquad \qquad \nabla := (1)\bar{e}_x + (1)\bar{e}_y \end{aligned} \tag{5.1.2}$$

Clairement, le gradient existe toujours et ne s'annule jamais. Donc, aucun point stationnaire à l'intérieur du disque.

Analysons la frontière. Le cercle peut être découpé en deux parties  $L_1 : y = \sqrt{1-x^2}$  et  $L_2 : y = -\sqrt{1-x^2}$  pour  $-1 < x < 1$ .

Il faudra pas oublier les extrémités  $x = \pm 1$  de ce découpage. Les points  $(\pm 1, 0)$  du domaine d'optimisation sont à retenir.

$$\begin{aligned} > \text{ValeursRetenues} := \{[1,0], [-1,0]\}; \\ \text{ValeursRetenues} := \{[-1,0], [1,0]\} \end{aligned} \tag{5.1.3}$$

Sur  $L_1$  où  $y = \sqrt{1-x^2}$  : analysons la fonction pour  $-1 < x < 1$ .

$$\begin{aligned} > f(x, \text{sqrt}(-x^2 + 1)); \\ & \qquad \qquad \qquad x + \sqrt{-x^2 + 1} + 22 \end{aligned} \tag{5.1.4}$$

Trouvons les valeurs critiques

$$\begin{aligned} > \text{solve}(\text{diff}((5.1.4), x) = 0, \{x\}); \\ & \qquad \qquad \qquad \left\{ x = \frac{\sqrt{2}}{2} \right\} \end{aligned} \tag{5.1.5}$$

$$\begin{aligned} > \text{eval}(\text{sqrt}(-x^2 + 1), x = \text{sqrt}(2)/2); \\ & \qquad \qquad \qquad \frac{\sqrt{2}}{2} \end{aligned} \tag{5.1.6}$$

Alors, sur  $L_1$ , on retient le point  $\left( \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right)$ .

```
> ValeursRetenues:=ValeursRetenues union {[sqrt(2)/2,sqrt(2)/2]};
      ValeursRetenues :=  $\left\{ [-1,0], [1,0], \left[ \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right] \right\}$  (5.1.7)
```

Sur  $L_2$  où  $y = -\sqrt{1-x^2}$  : analysons la fonction pour  $-1 < x < 1$ .

```
> f(x,-sqrt(-x^2 + 1));
       $x - \sqrt{-x^2 + 1} + 22$  (5.1.8)
```

Trouvons les valeurs critiques

```
> solve(diff((5.1.8),x)=0,{x});
       $\left\{ x = -\frac{\sqrt{2}}{2} \right\}$  (5.1.9)
```

```
> eval(-sqrt(-x^2 + 1),x=-sqrt(2)/2);
       $-\frac{\sqrt{2}}{2}$  (5.1.10)
```

Alors, sur  $L_2$ , on retient le point  $\left( -\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right)$ .

```
> ValeursRetenues:=ValeursRetenues union {[-sqrt(2)/2,-sqrt(2)/2]}
      ValeursRetenues :=  $\left\{ [-1,0], [1,0], \left[ -\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2} \right], \left[ \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right] \right\}$  (5.1.11)
```

Dans un tableau, comparons les valeurs de la fonction avec les points retenus. La notation d'ensemble a permis d'éviter les redondances.

```
> Valeurs_C:=op(ValeursRetenues);
      Valeurs_C :=  $[-1,0], [1,0], \left[ -\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2} \right], \left[ \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right]$  (5.1.12)
```

```
> printf("\n%18s \n","Tableau comparatif des valeurs de f aux points
retenus");
printf(`\n
| Approx. de f`);
printf(`\n
=====
=====`);
seq(printf(`\n %35a |%13a | %5g`, [Valeurs_C][k],f(op(1,
op(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),
evalf(f(op(1,op(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]
))))),k=1..nops([Valeurs_C]));
#seq(printf(`\n %35a | %10a | %5g`, [Valeurs_C][k],f(op(1,
op(k,[Valeurs_C])),op(2,op(k,[Valeurs_C]))),op(2,op(k,[Valeurs_C]
))),k=1..nops([Valeurs_C]));
```

Tableau comparatif des valeurs de f aux points retenus

Points retenus	Valeurs de f	Approx. de f
----------------	--------------	--------------




	[-1, 0]	21	21
	[1, 0]	23	23
[-1/2*2^(1/2), -1/2*2^(1/2)]		-2^(1/2)+22	20.5858
[1/2*2^(1/2), 1/2*2^(1/2)]		2^(1/2)+22	23.4142

En conclusion, le maximum absolu de  $f$  sur  $D$  est  $22 + \sqrt{2}$  et est atteint en  $\left(\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right)$ . Le minimum absolu est  $22 - \sqrt{2}$  et est atteint en  $\left(-\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right)$ .

Illustrons les résultats obtenus.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..2,y=-2..2,style=surface,grid=
[20,20]):
Points:=pointplot3d([[sqrt(2)/2,sqrt(2)/2,f(sqrt(2)/2,sqrt(2)/2)],
[-sqrt(2)/2,-sqrt(2)/2,f(-sqrt(2)/2,-sqrt(2)/2)]],symbol=
solidsphere,symbolsize=15,color="Niagara 2"):
Segment1:=pointplot3d([[sqrt(2)/2,sqrt(2)/2,0],[sqrt(2)/2,sqrt(2)
/2,f(sqrt(2)/2,sqrt(2)/2)]],style=line,thickness=1,
linestyle=3,
color="Niagara 2"):
Segment2:=spacecurve([[-sqrt(2)/2,-sqrt(2)/2,0],[-sqrt(2)/2,-sqrt
(2)/2,f(-sqrt(2)/2,-sqrt(2)/2)]],style=line,thickness=1,
linestyle=3,
color="Niagara 2"):
Projection1:=spacecurve([x,sqrt(1-x^2),f(x,sqrt(1-x^2))],x=-1..1,
color="Dalton 2",thickness=2):
Projection2:=spacecurve([x,-sqrt(1-x^2),f(x,-sqrt(1-x^2))],x=-1.
.1,color="Dalton 2",thickness=2):
display([Surface,Domaine_optimisation,Bord,Points,Segment||(1.
.2),Domaine],axes=framed,
Projection||(1..2),
shading=zhue,
orientation=[70,60,0]);
```



### 3.2 Méthode avec paramétrisation

Dans la solution précédente, le lecteur aura sans doute remarqué que le tracé de la frontière du domaine d'optimisation a été obtenu avec

```
Bord:=spacecurve( [cos(t),sin(t),0], t = 0..2π, color="Html 12", thickness=2):
```

Ce qui a permis de tracer, dans le plan  $xOy$ , un cercle de rayon unité centré en  $(0,0,0)$ . Au secondaire, lors de votre étude de la trigonométrie, vous avez sans doute étudié la fonction d'enroulement :

$$\begin{aligned} f: \mathbb{R} &\rightarrow [-1,1] \times [-1,1] \\ \theta &\mapsto (\cos \theta, \sin \theta). \end{aligned}$$

En fait, cette fonction correspond, dans le plan cartésien, à une description paramétrique d'un cercle centré à l'origine de rayon 1. Plus généralement, un cercle de rayon  $r$  centré en  $(a,b)$  peut être décrit, à l'aide d'une troisième variable  $t$ ,  $0 \leq t < 2\pi$ , par des équations paramétriques suivantes:

$$\begin{aligned} x &= a + r \cos(t) \\ y &= b + r \sin(t) \end{aligned}$$

Avec la paramétrisation  $x = \cos(t)$ ,  $y = \sin(t)$ ,  $0 < t < 2\pi$ , la fonction à optimiser sur le disque devient

$$\left[ \begin{array}{l} > \mathbf{f(\cos(t), \sin(t))}; \\ \qquad \qquad \qquad \cos(t) + \sin(t) + 22 \end{array} \right. \quad (5.2.1)$$

Déterminons si  $f$  possède des points critiques à l'intérieur (sur l'intervalle ouvert)  $0 < t < 2\pi$

$$\left[ \begin{array}{l} > \mathbf{diff(f(\cos(t), \sin(t)), t)}; \\ \qquad \qquad \qquad -\sin(t) + \cos(t) \end{array} \right. \quad (5.2.2)$$

$$\left[ \begin{array}{l} > \mathbf{solve([(5.2.2)=0, 0 < t, t < 2*Pi], t, allsolutions)}; \\ \qquad \qquad \qquad \left\{ t = \frac{1}{4} \pi + \pi\_Z1 \right\} \end{array} \right. \quad (5.2.3)$$

```
> about(_Z1);
Originally _Z1, renamed _Z1~:
is assumed to be: AndProp(integer,RealRange(0,1))
```

Sur la frontière, il y a donc les points

```
> subs(t=Pi/4,[cos(t),sin(t)]=eval(subs(t=Pi/4,[cos(t),sin(t)]));
subs(t=5*Pi/4,[cos(t),sin(t)]=eval(subs(t=5*Pi/4,[cos(t),sin(t)]))
);
```

$$\left[ \cos\left(\frac{\pi}{4}\right), \sin\left(\frac{\pi}{4}\right) \right] = \left[ \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right]$$

$$\left[ \cos\left(\frac{5\pi}{4}\right), \sin\left(\frac{5\pi}{4}\right) \right] = \left[ -\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2} \right] \quad (5.2.4)$$

Les points  $\left(\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right)$  et  $\left(-\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right)$  sont à retenir. N'oublions pas " l'extrémité " de la frontière circulaire, soit le point  $(1, 0)$ , qui correspond à  $t = 0$ .

Nous avons déjà montré que  $f$  ne possédait pas de point stationnaire à retenir.

### Tableau comparatif

aucun pt stationnaire	$t = \frac{\pi}{4}$	$t = \frac{5\pi}{4}$	$t = 0$
	<pre>&gt; f(cos(pi/4), sin(pi/4)) sqrt(2) + 22 &gt; evalf(%); 23.41421</pre>	<pre>&gt; f(cos(5*pi/4), sin(5*pi/4)) -sqrt(2) + 22 &gt; evalf(%); 20.58579</pre>	<pre>&gt; f(cos(0), sin(0)) 23</pre>

Même conclusion mais obtenue plus rapidement.

## 3.3 Méthode avec les multiplicateurs de Lagrange

Rappelons que nous avons à optimiser  $f(x, y) = x + y + 22$  sur le disque  $x^2 + y^2 \leq 1$

```
> f:=(x,y)->x+y+22;
f := (x,y) ↦ x + y + 22 (5.3.1)
```

Rappelons aussi que nous avons déjà montré qu'il n'y avait aucun candidat à l'intérieur du domaine d'optimisation à retenir. Nous allons appliquer la méthode des multiplicateurs de Lagrange pour rechercher des candidats à retenir sur la frontière du domaine d'optimisation, soit le disque  $x^2 + y^2 \leq 1$ .

La contrainte  $g(x, y) = 0$  dans la méthode des multiplicateurs de Lagrange sera donc l'équation du cercle

unité.

$$\left[ \begin{array}{l} > \mathbf{g}:=(\mathbf{x},\mathbf{y})\rightarrow\mathbf{x}^2+\mathbf{y}^2-1; \\ & \qquad \qquad \qquad g := (x, y) \mapsto x^2 + y^2 - 1 \end{array} \right. \quad (5.3.2)$$

Formons maintenant la fonction  $F$  définie par  $F(x, y, \lambda) = f(x, y) + \lambda g(x, y)$ . Le paramètre «  $\lambda$  » est appelé multiplicateur de Lagrange.

$$\left[ \begin{array}{l} > \mathbf{F}:=(\mathbf{x},\mathbf{y},\mathbf{lambda})\rightarrow\mathbf{f}(\mathbf{x},\mathbf{y})+\mathbf{lambda}*\mathbf{g}(\mathbf{x},\mathbf{y}); \\ & \qquad \qquad \qquad F := (x, y, \lambda) \mapsto f(x, y) + \lambda \cdot g(x, y) \end{array} \right. \quad (5.3.3)$$

Recherchons maintenant les solutions de  $\nabla f(x, y, \lambda) = (0, 0, 0)$ .

$$\left[ \begin{array}{l} > \mathbf{Nabla}:=\mathbf{Gradient}(\mathbf{F}(\mathbf{x},\mathbf{y},\mathbf{lambda}),[\mathbf{x},\mathbf{y},\mathbf{lambda}]); \\ & \qquad \qquad \qquad \nabla := (2\lambda x + 1)\bar{e}_x + (2\lambda y + 1)\bar{e}_y + (x^2 + y^2 - 1)\bar{e}_\lambda \end{array} \right. \quad (5.3.4)$$

$$\left[ \begin{array}{l} > \mathbf{Sol}:=\mathbf{solve}(\{\mathbf{Nabla}[1]=0,\mathbf{Nabla}[2]=0,\mathbf{Nabla}[3]=0\},\{\mathbf{x},\mathbf{y},\mathbf{lambda}\}, \\ \quad \mathbf{explicit}); \\ & \qquad \qquad \qquad \mathit{Sol} := \left\{ \lambda = -\frac{\sqrt{2}}{2}, x = \frac{\sqrt{2}}{2}, y = \frac{\sqrt{2}}{2} \right\}, \left\{ \lambda = \frac{\sqrt{2}}{2}, x = -\frac{\sqrt{2}}{2}, y = -\frac{\sqrt{2}}{2} \right\} \end{array} \right. \quad (5.3.5)$$

Donc, aucun point intérieur à retenir et, ici, deux seuls points à retenir sur la frontière du disque:

$\left( \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right)$  et  $\left( -\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2} \right)$ . Calculons leur image;

$$\left[ \begin{array}{l} > \mathbf{f}'(\mathbf{sqrt}(2)/2,\mathbf{sqrt}(2)/2)=\mathbf{f}(\mathbf{sqrt}(2)/2,\mathbf{sqrt}(2)/2); \\ & \mathbf{f}'(-\mathbf{sqrt}(2)/2,-\mathbf{sqrt}(2)/2)=\mathbf{f}(-\mathbf{sqrt}(2)/2,-\mathbf{sqrt}(2)/2); \\ & \qquad \qquad \qquad f\left(\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right) = \sqrt{2} + 22 \\ & \qquad \qquad \qquad f\left(-\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right) = -\sqrt{2} + 22 \end{array} \right. \quad (5.3.6)$$

Sur le disque, le maximum de  $f$  vaut  $22 + \sqrt{22}$  atteint en  $\left( \frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2} \right)$  et le minimum  $22 - \sqrt{22}$  atteint en  $\left( -\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2} \right)$ .


Même conclusion.

$$\left[ \begin{array}{l} > \mathbf{Contrainte}:=\mathbf{seq}(\mathbf{spacecurve}([\mathbf{sin}(t),\mathbf{cos}(t),\mathbf{k}],t=0..2*\mathbf{Pi},\mathbf{color}=\mathbf{"Niagara 4"},\mathbf{thickness}=1),\mathbf{k}=[2,4,6,8,10,12,14,16,18]): \\ & \mathbf{Contrainte\_S}:=\mathbf{spacecurve}([\mathbf{sin}(t),\mathbf{cos}(t),\mathbf{f}(\mathbf{sin}(t),\mathbf{cos}(t))],t=0..2*\mathbf{Pi},\mathbf{color}=\mathbf{"Niagara 1"},\mathbf{thickness}=2): \\ & \mathbf{Bord}:=\mathbf{spacecurve}([\mathbf{cos}(t),\mathbf{sin}(t),0],t=0..2*\mathbf{Pi},\mathbf{color}=\mathbf{"Dalton 2"},\mathbf{thickness}=2): \\ & \mathbf{BordSup}:=\mathbf{spacecurve}([\mathbf{cos}(t),\mathbf{sin}(t),\mathbf{f}(\mathbf{cos}(t),\mathbf{sin}(t))],t=0..2*\mathbf{Pi},\mathbf{color}=\mathbf{"Dalton 2"},\mathbf{thickness}=2): \\ & \mathbf{L}:=\mathbf{seq}(\mathbf{plots}[\mathbf{spacecurve}]([\mathbf{cos}(k*\mathbf{Pi}/36),\mathbf{sin}(k*\mathbf{Pi}/36),0],[\mathbf{cos}(k*$$

```

Pi/36),sin(k*Pi/36),f(cos(k*Pi/36),sin(k*Pi/36))]],
color="Niagara 4",thickness=0),k=0..72):
A:=display([Domaine,Domaine_optimisation,Surface,Contrainte,
Contrainte_S,L,Bord],style=patchnogrid):
B:=display([Domaine,Domaine_optimisation,Surface,Bord,BordSup,
Points],style=patchnogrid,Segment| |($1..2)):
display(Matrix(1,2,[A,B]),orientation=[70,60,0],axes=framed);

```



## Exemple 4

Soit à optimiser  $f(x,y) = xy(x+y-3) + 1$  sur le domaine  $D = \{(x,y) \mid 0 \leq x \leq y(3-2y)\}$ .

Dans le plan cartésien, l'équation  $x = y(3-2y)$  est une parabole horizontale ouverte vers la gauche dont l'abscisse du sommet est  $-\frac{b}{2a} = \frac{3}{4}$ . Les racines sont donc  $y = 0$  et  $y = \frac{3}{2}$ .

```
> f:=(x,y)->x*y*(x+y-3)+3/2;
```


$$f := (x,y) \mapsto x \cdot y \cdot (x+y-3) + \frac{3}{2}$$

(6.1)

```

> Surface:=plot3d([x,y,f(x,y)],x=0..1.55,y=0..1.55,transparency=0.25,
grid=[20,20]):
Domaine:=plot3d([x,y,0],x=0..1.5,y=0..1.5,grid=[20,20],style=
wireframe,color="Niagara 16"):
Domaine_optimisation:=plot3d([x,y,0],x=0..y*(3-2*y),y=0..3/2,
style=patchnogrid,color="Dalton 7"):
display([Surface,Domaine,Domaine_optimisation],axes=framed,
orientation=[-40,70,0]);

```



Déterminons si  $f$  possède des points critiques à l'intérieur du domaine d'optimisation.

$$\begin{aligned} > \text{Nabla} := \text{Gradient}(f(x, y), [x, y]); \\ \nabla := (y(x + y - 3) + xy)\bar{e}_x + (x(x + y - 3) + xy)\bar{e}_y \end{aligned} \quad (6.2)$$

$$\begin{aligned} > \text{solve}(\{\text{Nabla}[1]=0, \text{Nabla}[2]=0, x>0, y>0\}, \{x, y\}); \\ \{x=1, y=1\} \end{aligned} \quad (6.3)$$

Attention ! Le point  $(1, 1)$  n'est pas à retenir car ce n'est pas un point intérieur mais un point sur la frontière.

En effet

$$\begin{aligned} > \text{Eval}(y*(3 - 2*y), y=1) = \text{eval}(y*(3 - 2*y), y=1); \\ y(3 - 2y) \Big|_{y=1} = 1 \end{aligned} \quad (6.4)$$

Analysons les frontières. Il y en a deux.

Sur  $L_1$  où  $x=0$  et  $0 < y < \frac{3}{2}$ .

$$\begin{aligned} > f(0, y); \\ \frac{3}{2} \end{aligned} \quad (6.5)$$

Ce qui montre que sur  $L_1$ , la fonction  $f$  est constante (dérivée nulle) et donc tous ces points  $(0, y)$  sont des points critiques.

Sur  $L_2$  (segment de la parabole) où  $x = y(3 - 2y)$  pour  $0 < y < \frac{3}{2}$ ,

$$\begin{aligned} > f(y*(3-2*y), y); \\ y^2(3 - 2y)(y(3 - 2y) + y - 3) + \frac{3}{2} \end{aligned} \quad (6.6)$$

Trouvons les valeurs critiques

$$\begin{aligned} > \text{solve}(\{\text{diff}((6.6), y)=0, y>0, y<3/2\}, \{y\}); \\ \{y=1\} \end{aligned} \quad (6.7)$$

On retrouve le point (1,1) ! Qui est vraiment sur la frontière où f prend la valeur  $\frac{1}{2}$ .

$$f(1,1);$$

$$\frac{1}{2}$$

(6.8)

Alors, sur  $L_2$ , on retient le point (1, 1).

Reste à comparer les candidats trouvés en n'oubliant pas les coins (0, 0) et  $(\frac{3}{2}, 0)$ .

### Tableau comparatif


aucun pt stationnaire	$(0,y)$ $0 < y < \frac{3}{2}$	(1,1)	(0,0)	$(\frac{3}{2}, 0)$
	$\frac{3}{2}$	$f(1,1);$ $\frac{1}{2}$	$f(0,0);$ $\frac{3}{2}$	$f(3/2,0);$ $\frac{3}{2}$

En conclusion, le maximum absolu de f sur D est  $\frac{3}{2}$  et est atteint en tous les points (0, y) où  $0 \leq y \leq \frac{3}{2}$ . Le minimum absolu est  $\frac{1}{2}$  et est atteint en (1, 1).

Illustrons les résultats obtenus.

```
> Surface:=plot3d([x,y,f(x,y)],x=0..1.55,y=0..1.55,color="Nautical 14",
lightmodel=light4,style=surface):
Domaine:=plot3d([x,y,0],x=0..1.5,y=0..1.5,grid=[25,25],style=
wireframe,color="Niagara 16"):
Domaine_optimisation:=plot3d([x,y,0],x=0..y*(3-2*y), y=0..3/2,style=
patchnograd,color="Dalton 4"):
ProjectionL1:=spacecurve([-2*(y - 3/4)^2 + 9/8,y,f(-2*(y - 3/4)^2 +
9/8,y)],y=0..3/2,color="Dalton 4",thickness=2):
ProjectionL2:=pointplot3d([[0,0,f(0,0)],[0,3/2,f(0,3/2)]],style=line,
color="Dalton 7"):
Segment_min:=pointplot3d([[1,1,0],[1,1,f(1,1)]],style=line,color=
"Niagara 2",thickness=1,linestyle=3):
Pas:=[seq(0+k*3/20,k=0..10)]:
Segments:=seq(spacecurve([[0,k,0],[0,k,f(0,k)]]),color="Niagara 2",
thickness=0,linestyle=3),k=Pas):
Points:=pointplot3d([[1,1,f(1,1)],seq([0,k,f(0,k)],k=Pas)],symbol=
solidsphere,symbolsize=15,color="Niagara 2"):
Lieu1:=spacecurve([1-t,0,f(1-t,0)+.01],t=0..1,color=navy,thickness=3)
:
Lieu2:=spacecurve([0,1-t,f(0,1-t)+.01],t=0..1,color=navy,thickness=3)
:
Lieu3:=spacecurve([t,1-t,f(t,1-t)],t=0..1,color=navy,thickness=3):
```

```
A:=display([Domaine,ProjectionL2,Points,Domaine_optimisation,
Segment_min,Segments,ProjectionL1,Surface],orientation=[-35,75,0]):
B:=display([Domaine,ProjectionL2,ProjectionL1,Points,
Domaine_optimisation,Segment_min,Segments],orientation=[55,75,0]):
display(Matrix(1,2,[A,B]),axes=framed);
```



## Exemple 5

Optimiser  $f(x, y) = 3y - 2x + 10$  sur le domaine  $D = \{(x, y) \mid x \geq 0, y \geq 0, y \leq x + 1, y \leq 4 - 2x\}$ .

```
> f:=(x,y)->3*y-2*x+10;
      f := (x,y) ↦ 3·y - 2·x + 10 (7.1)
```

La fonction  $f$  est de la forme  $Ax + By + C$ . Si  $A \neq 0 \vee B \neq 0$ , il est clair que  $\nabla f(x, y) \neq \langle 0, 0 \rangle$  et donc que  $f$  ne peut avoir d'extrema absolus à l'intérieur de son domaine d'optimisation.

```
> Nabla:=convert(Gradient(f(x,y),[x,y]),Vector[row]);
      ∇ := [ -2  3 ] (7.2)
```


On le savait !

Les contraintes de non-négativité de  $x$  et de  $y$  permettent de déduire que  $0 \leq x \leq 2$ .

```
> Surface:=plot3d([x,y,f(x,y)],x=0..2,y=0..2,grid=[15,15],shading=zhue)
:
Domaine:=plot3d([x,y,0],x=0..2,y=0..2,grid=[15,15],style=wireframe,
color="Niagara 16"):
Domaine_optimisation:=plot3d([x,y,0],x=0..2,y=0..min(x+1,4-2*x),
style=patchnogrid,color="MapleV 1"):
Frontière:=polygonplot3d([[0,0,0],[2,0,0],[1,2,0],[0,1,0],[0,0,0]],
```



```
color="HTML 26",thickness=2):
display([Surface,Domaine,Domaine_optimisation,Frontière],axes=normal,
orientation=[25,65,0]);
```



Il nous faut d'abord trouver le point d'intersection des droites  $y = x + 1$  et  $y = 4 - 2x$  afin de faire l'analyse de  $f$  sur le bord.

```
> solve({y=x+1,y=4-2*x},{x,y});
{x=1,y=2} (7.3)
```

(Maple a simplement été utilisé comme exemple de résolution...)

Reste à analyser  $f$  sur les frontières. Il y en a quatre.

Sur chaque frontière, nous allons retenir automatiquement les extrémités.

$L_1$  où  $x = 0$ . Analysons la fonction pour  $0 < y < 1$ .

```
> f(0,y);
3y + 10 (7.4)
```

La dérivée ne s'annule jamais, on retient seulement les extrémités: (0,0) et (0,1).

$L_2$  où  $y = 0$ . Analysons la fonction pour  $0 < x < 1$ .

```
> f(x,0);
10 - 2x (7.5)
```

La dérivée ne s'annule jamais, on retient seulement les extrémités: (0,0) et (1,0).

$L_3$  où  $y = x + 1$ . Analysons la fonction pour  $0 < x < 1$ .

```
> f(x,x+1);
x + 13 (7.6)
```

La dérivée ne s'annule jamais, on retient seulement les extrémités: (0,1) et (1,2).

$L_4$  où  $y = 4 - 2x$ . Analysons la fonction pour  $1 < x < 2$ .

```
> f(x,4-2*x);
```

$$22 - 8x$$

(7.7)

La dérivée ne s'annule jamais, on retient seulement les extrémités: (1,2) et (2,0).

Dans un tableau, comparons les valeurs de la fonction avec les points retenus, tout en évitant les redondances.


### Tableau comparatif

aucun pt stationnaire	(0,0)	(0,1)	(1,0)	(1,2)	(2,0)
	> $f(0,0);$ 10	> $f(0,1);$ 13	> $f(1,0);$ 8	> $f(1,2);$ 14	> $f(2,0);$ 6

En conclusion, le maximum absolu de  $f$  sur  $D$  est 14 et est atteint en (1,2) (coin). Le minimum absolu est 6 et est atteint en (2,0) (coin)

Illustrons les résultats obtenus.

```
> PolygoneSup:=polygonplot3d([[0,0,f(0,0)],[2,0,f(2,0)],[1,2,f(1,2)],[0,1,f(0,1)],[0,0,f(0,0)]],color="MapleV 5");
Segment1:=pointplot3d([[0,0,0],[0,0,f(0,0)]],style=line,color="Niagara 1",thickness=1,linestyle=2):
Segment2:=pointplot3d([[2,0,0],[2,0,f(2,0)]],style=line,color="Niagara 1",thickness=1,linestyle=2):
Segment3:=pointplot3d([[1,2,0],[1,2,f(1,2)]],style=line,color="Niagara 1",thickness=1,linestyle=2):
Segment4:=pointplot3d([[0,1,0],[0,1,f(0,1)]],style=line,color="Niagara 1",thickness=1,linestyle=2):
SegmentMin:=pointplot3d([[2,0,0],[2,0,f(2,0)]],style=line,color="Niagara 2",thickness=1,linestyle=3):
SegmentMax:=pointplot3d([[1,2,0],[1,2,f(1,2)]],style=line,color="Niagara 2",thickness=1,linestyle=3):
A:=display([PolygoneSup,Domaine,Domaine_optimisation,Surface,Segment|| (1..4)],axes=framed,orientation=[60,55,0]):
Points:=pointplot3d([[1,2,f(1,2)],[2,0,f(2,0)]],symbol=solidSphere,symbolsize=20,color="Niagara 2"):
B:=display([PolygoneSup,Domaine,Domaine_optimisation,Points,SegmentMin,SegmentMax,Segment|| (1..4)],axes=normal,orientation=[-10,75,0]):
display(Matrix(1,2,[A,B]));
```



Les inégalités composant le domaine d'optimisation étant des inégalités linéaires, elle forment ce qu'on appelle, en *programmation linéaire*, le polygone des solutions réalisables. Sans le démontrer ici, on peut affirmer que les extrema globaux se trouvent toujours sur les arêtes du polygone des solutions réalisables (le bord du domaine d'optimisation), mais toujours sur les "coins".

```
> with(simplex,maximize,minimize);
                                [maximize,minimize] (7.8)
```

```
> FonctionObjective:=f(x,y);
   Contraintes:={y-x<=1,y+2*x<=4};
                   FonctionObjective := 3y - 2x + 10
                   Contraintes := {y - x ≤ 1, y + 2x ≤ 4} (7.9)
```

```
> maximize(FonctionObjective,Contraintes, NONNEGATIVE);
   minimize(FonctionObjective,Contraintes, NONNEGATIVE);
                   {x = 1, y = 2}
                   {x = 2, y = 0} (7.10)
```

```
> Max:=f(1,2);
   Min:=f(2,0);
                   Max := 14
                   Min := 6 (7.11)
```